


**Решение заданий олимпиады  
«Будущее Кузбасса. Информатика и ИКТ»**  
(задания представлены в порядке возрастания сложности  
и, соответственно, баллов за их правильное решение)

**Задание 1. Максимальная оценка – 5 баллов.**

Дан фрагмент электронной таблицы:

	А	В	С
1	2		=A1+1
2	=C1-B1	=(3*B1+C1)/3	=B2+A1



Какое число должно быть записано в ячейке В1, чтобы построенная после выполнения вычислений диаграмма по значениям диапазона ячеек А2:С2 соответствовала рисунку? Известно, что все значения диапазона, по которым построена диаграмма, имеют один и тот же знак.

**Решение:**

В ячейке С1 будет значение 3.  $C1 = 3$ ,  $A1 = 2$ , следовательно,  $B2 = B1 + 1$ ,  $C2 = B2 + 2 = B1 + 3$ . Следовательно,  $B2$  не равно  $C2$ , то есть, одно из них (судя по диаграмме) больше другого в 2 раза. Решим эти уравнения последовательно для  $B2 = 2C2$  и  $C2 = 2B2$ . В первом случае  $B1 = -5$ ,  $A2 = 8$ ,  $B2 = -4$ ,  $C2 = -2$ . Это не соответствует диаграмме. Во втором случае  $B1 = 1$ ,  $A2 = B2 = 2$ ,  $C2 = 4$ . Этот вариант соответствует диаграмме, следовательно, правильный ответ – 1.

**Ответ: 1**

**Задание 2. Максимальная оценка – 5 баллов.**

Запишите значение переменной  $s$ , полученное в результате работы следующей программы. Текст программы приведён на трёх языках программирования.

Алгоритмический язык	Бейсик	Паскаль
<pre> алг нач цел s, k  s := 0 нц для k от 6 до 12     s := s+10 кц вывод s кон                     </pre>	<pre> DIM k, s AS INTEGER  s = 0 FOR k = 6 TO 12     s = s+10 NEXT k PRINT s                     </pre>	<pre> Var s, k: integer;  Begin s := 0; for k := 6 to 12 do     s := s+10; writeln(s); End.                     </pre>

**Ответ: 70**

**Задание 3. Максимальная оценка – 10 баллов.**

В символьной строке имеются круглые скобки. Проверить, правильно ли они расставлены.

**Пример решения:**

Обозначим  $K$  - число левых скобок,  $M$  - число правых скобок, тогда, на каждом шаге подсчета скобок, должно выполняться условие:  $K \geq M$ . После подсчета всех скобок должно выполняться условие  $K = M$ .

Необходимо написать программу на любом языке программирования для решения поставленной задачи. Перед программой укажите версию языка программирования.

**Задание 4. Максимальная оценка – 10 баллов.**

На обработку поступает положительное целое число, не превышающее  $10^9$ . Необходимо написать программу на любом языке программирования, которая выводит на экран количество разрядов числа, имеющих четные значения. Перед программой укажите версию языка программирования.

**Пример решения:**

```
var N, digit, sum: longint;
begin
  readln(N);
  sum := 0;
  while N > 0 do
  begin
    digit := N mod 10;
    if digit mod 2 = 0 then
      sum := sum + 1;
    N := N div 10;
  end;
  writeln(sum)
end.
```

**Задание 5. Максимальная оценка – 10 баллов.**

Распечатать числовую последовательность, которая задается по следующим правилам:

- первое число последовательности - натуральное число, кратное 3 (входной параметр задачи);
- каждый последующий элемент равен сумме кубов цифр предыдущего.

Например:

```
33
33+33=54
53+43=189
13+83+93=1242
13+23+43+23=81
83+13=513
53+13+33=153
```

Вычисления прекратить, когда очередной элемент последовательности станет равен 153. (Известно, что любая такая последовательность рано или поздно приводит к 153).

**Пример решения:**

На каждом шаге данного алгоритма приходится разбивать целое число на отдельные цифры (причем количество цифр в числе неизвестно). Это можно выполнить, используя операции целочисленной арифметики (деления нацело - `div` и остатка от деления - `mod`). Процесс вычисления очередного члена последовательности  $p$  через предыдущий в рассматриваемой задаче будет иметь следующий вид ( $s$  и  $p1$  - рабочие переменные,  $t$  - очередная цифра числа). Перед программой укажите версию языка программирования.

**Pascal:**

```
s:=0;
p1:=p;
while p1<>0 do
begin
  t:=p1 mod 10;
  p1:=p1 div 10;
  s:=s+t*t*t;
end;
p:=s
```

**Задание 6. Максимальная оценка – 15 баллов.**

Поднимаясь по лестнице, заяц прыгает либо на следующую ступеньку, либо через одну, либо через две ступеньки. Сколькими способами он может подняться на ступеньку с номером  $n$ ?

**Пример решения:**

```
Uses Crt;
Var n , i : 1..36;
a , b , c , q : LongInt;
BEGIN
Write('Введите количество ступенек (от 1 до 36) ');
ReadLn(n);
a:=1;
b:=2;
c:=4;
Case n of
  2: q:=2;
  3: q:=4;
else
for i:=n downto 4 do
begin
  q:=a+b+c;
  a:=b; b:=c; c:=q;
end;
end; {case}
Write('Всего способов: ',q); ReadLn
END.
```

**Задание 7. Максимальная оценка – 20 баллов.**

Разложить натуральное число на простые сомножители. Необходимо написать программу на любом языке программирования для решения поставленной задачи. Перед программой укажите версию языка программирования.

**Пример решения:**

```
uses crt;
procedure PrsDel(a:longint);
var n:Longint;
    b:boolean;
begin
n:=2;
while a>0 do
begin
  b:=true;
  while n <= round(sqrt(a)) do
  begin
    if a mod n=0 then
```

```

begin
  write(n, ' ');
  a:=a div n;
  b:=false;
  break;
end;
inc(n);
end;
if b then
begin
  write(a);
  break;
end;
end;
end;
var a:longint;
begin
clrscr;
write('a=');readln(a);
PrsDel(a);
readln
end.

```

### **Задание 8. Максимальная оценка – 25 баллов**

Специальная камера, установленная на перекрёстке, фиксирует количество проезжающих автомобилей, и каждую минуту по каналу связи передаёт неотрицательное целое число — количество автомобилей, проехавших перекрёсток за эту минуту. Известно, что за минуту перекрёсток может проехать не более 100 автомобилей. Необходимо найти в заданной серии показаний максимальное количество автомобилей, проехавших перекрёсток в течение пяти подряд идущих минут. Максимальное количество показаний, которое может передать камера, не превышает 1440.

Напишите на любом языке программирования программу для решения поставленной задачи. Для получения максимального результата программа должна быть эффективна по времени и по используемой памяти.

Входные данные представлены следующим образом. В первой строке задаётся число  $N$  – общее количество переданных показаний. Гарантируется, что  $N > 5$ . В каждой из следующих  $N$  строк задаётся одно положительное целое число – очередное показание камеры.

Пример входных данных:

8; 5; 12; 27; 10; 4; 50; 7;16

Программа выводит только одно число – наибольшее количество автомобилей, проехавших перекрёсток за пять подряд идущих минут.

Пример выходных данных для приведённого выше примера входных данных:  
103

### **Пример решения на 15 баллов:**

Первый вариант решения заключается в том, чтобы объявить массив с максимально возможным количеством элементов – 1440:

```
var a: array[1..1440] of byte;
```

Далее вводится значение  $N$  и считаются данные в массив:

```
for i:=1 to N do
  readln(a[i]);
```

Теперь с помощью вложенных циклов (один цикл внутри другого) будем находить сумму пяти текущих элементов, и сравнивать её с максимальным значением. При этом каждый повтор первого цикла переменную  $S$  будем обнулять:

```
for i:=1 to N-4 do
begin
  for j:=0 to 4 do
    s := s+a[i+j];
    if s > max then
      max := s;
    s := 0;
end;
```

Выводим значение переменной  $max$ :

```
writeln(max);
```

### Полное решение:

```
var
a: array[1..1440] of byte;
i, j, N, max, s: integer;
begin
  readln(N);
  for i:=1 to N do
    readln(a[i]);
  max := 0; s := 0;
  for i:=1 to N-4 do
  begin
    for j:=0 to 4 do
      s := s+a[i+j];
      if s > max then
        max := s;
      s := 0;
    end;
  writeln(max);
end.
```

Данная программа не является эффективной по памяти, так как хранит все введенные данные.

### Пример решения на 25 баллов:

Для решения задания мы можем не использовать такой огромный массив, достаточно хранить в памяти всего пять значений, сумму которых мы находим:

```
var a: array[1..5] of byte;
```

Сначала введем  $N$  и считаем в массив первые пять значений, при этом сразу можно найти и их сумму:

```
readln(n);
for i:=1 to 5 do
begin
  readln(a[i]);
  s := s+a[i];
end;
```

Сразу же присвоим переменной  $max$  значение  $s$ :

```
max := s;
```

Теперь введём оставшиеся значения. Для этого будем использовать последний (пятый) элемент массива, предварительно сдвинув все элементы на один влево:

```
for i:=6 to N do
begin
```

```

for j:=1 to 4 do //сдвиг массива
  a[j] := a[j+1];
readln(a[5]);

```

Первое, что приходит в голову – просто посчитать сумму элементов с помощью вложенного цикла, но можно поступить лучше. Сумму элементов мы можем посчитать, вычтя из предыдущей суммы первый элемент до сдвига, и прибавив к ней введённый элемент после сдвига:

5 7 1 4 2 7 Сумма равна 26

Следующее значение

5 7 1 4 2 7 4 Сумма равна 26-5+4=25

То есть теперь код будет выглядеть так:

```

for i:=6 to N do
begin
  s := s - a[1];
  for j:=1 to 4 do
    a[j] := a[j+1];
  readln(a[5]);
  s := s + a[5]

```

Остаётся сравнить значение переменной *s* с *max*:

```

if s > max then    max :=s;

```

**Полное решение:**

```

var
a: array[1..5] of integer; s, i, j, max, N: integer;
begin
  readln(n);
  s := 0;
  for i:=1 to 5 do
  begin
    readln(a[i]);
    s := s+a[i];
  end;
  max := s;
  for i:=6 to N do
  begin
    s := s - a[1];
    for j:=1 to 4 do
      a[j] := a[j+1];
    readln(a[5]);
    s := s + a[5];
    if s > max then
      max :=s;
  end;
  writeln(max); end.

```